The Traffic Assignment Problem

1. Format of Dataset

1.1 Network File

The file is saved as an Excel file, using the following format:

1) First lines are **metadata** including:
   "Number of zones", "Number of nodes", "First thru node", "Number of links" (See Fig. 1).
   Note: For "First thru node", in some networks (like Sioux-Falls) it is equal to 1, indicating that traffic can move through all nodes, including centrals. In other networks when traffic is not allowed to go through zones, the zones are numbered 1 to n and the "First thru node" is set to n+1.
2) The other lines are **properties of links**, which including the following attributes by sequence:
   "Start node", "End node", "Capacity", , "Free-flow time", "B", "Power".
   Note: The name of attributes above should be replaced by "1", "2", "3", "4", "5", "6" (See Fig. 1). The number of each link should be listed in the first column. The link travel time function is a BPR function:

$$\text{Link travel time} = \text{Free flow time} \times \left(1 + \text{B} \times \left(\frac{\text{Flow}}{\text{Capacity}}\right)^{\text{Power}}\right)$$

   Note: Besides "Braess_net", "NineNodes_net", "Sioux-Falls_net", "Hull_net" and "NewHull_net" networks, the "Free-flow time" of the other networks has been divided by 100000 in case the variables will exceed the limit of value in GAMS (default 1.e10).

1.2 Trip Table

The table includs the demands of OD pairs, whose first column is origin ID, and the first row is the destination ID (See Fig. 1).

1.3 Input Format

The network file and trip table should be in the same sheet (like "Sheet1").

1.4 Output Format

The results are output in a LST file including the flow on each link, the cost of UE and SO.

2. Parameters in GAMS Programming

The GAMS codes are shown in Appendix 1.

1) The "set of links" is from 1 to "Number of links", i.e. /1*Number of links/.
2) The "set of parameters" represents the attributes of links, which is /1*6/.
3) The "set of nodes" is from 1 to "Number of nodes", i.e. /1*Number of nodes/.
4) The "set of origins" is from 1 to "Number of zones", i.e. /1*Number of zones /.
5) The "set of destinations" is from 1 to "Number of zones", i.e. /1*Number of zones /.

6) The “FirstNode” means the “First thru node”, which is equal to the number of “First thru node”.
7) When the datasets are input from Excel to GAMS, the following codes are utilized:
   $CALL GDXXRW.EXE “Name of Excel” par=data rng=A6:G24   par=od_data rng=A26:E30
   $GDXIN “Name of GDX”
   $LOAD data od_data
   $GDXIN
   Note:
   A6 is the coordinate in Excel where the properties of links begin to be listed;
   G24 is the coordinate where the properties of links end;
   A26 is the coordinate where the trip table begins;
   E30 is the coordinate where the trip table demand ends (See Fig. 1).



	A	B	C	D	E	F	G	H
1		Number of zones		4				
2		Number of nodes		9				
3		First thru node		1				
4		Number of links		18				
5								
6		1	2	3	4	5	6	
7	1	1	5	12	5	0.15	4	
8	2	1	6	18	6	0.15	4	
9	3	2	5	35	3	0.15	4	
10	4	2	6	35	9	0.15	4	
11	5	5	6	20	9	0.15	4	
12	6	5	7	11	2	0.15	4	
13	7	5	9	26	8	0.15	4	
14	8	6	5	11	4	0.15	4	
15	9	6	8	33	6	0.15	4	
16	10	6	9	32	7	0.15	4	
17	11	7	3	25	3	0.15	4	
18	12	7	4	24	6	0.15	4	
19	13	7	8	19	2	0.15	4	
20	14	8	3	39	8	0.15	4	
21	15	8	4	43	6	0.15	4	
22	16	8	7	36	4	0.15	4	
23	17	9	7	26	4	0.15	4	
24	18	9	8	30	8	0.15	4	
25								
26		1	2	3	4			
27	1	0	0	10	20			
28	2	0	0	30	40			
29	3	0	0	0	0			
30	4	0	0	0	0			
31								

Figure 1: Dataset Format

3. Result

Network	Zones	Nodes	Links	First thru node	UE	SO	(UE-SO)/UE	UE resource usage (sec)	SO resource usage (sec)
Braess	2	4	5	1	552.00	498.00	9.78%	0.02	0.00
NineNodes	4	9	18	1	2455.87	2253.92	8.22%	0.02	0.00
Sioux-Falls	24	24	76	1	74.80	71.94	3.82%	0.39	0.11
Hull	23	501	798	23	40053.34	37688.65	5.90%	15.29	1.97
NewHull	23	501	798	23	51350.74	50542.64	1.57%	12.20	4.18
Winnipeg	147	1052	2836	148	10.79	9.66	10.52%	13588.98	5241.56
Anaheim	38	416	914	39	14.95	14.67	1.89%	31.06	17.15
Barcelona	110	1020	2522	111	13.81	13.58	1.67%	1704.51	1010.23
Terrassa	55	1609	3264	56	77041.08	65978.86	14.36%	45263.07	25880.68
Chicago Sketch	387	933	2950	388	219.70	163.86	25.42%	193425.20	98201.87
Hessen	245	4660	6674	246					
Austin	7388	7388	18961	1					
Chicago Regional	1790	12982	39018	1791					
Philadelphia	1525	13389	40003	1526					
PRISM-M	898	14639	33937	899					

Table 1: The UE and SO results of different networks

Note:

1) The initial link flows of SO are the flow destination of UE.
2) Computer configuration: Intel(R) Core™ i5-3570 CPU @ 3.40GHz, 8.00 GB (7.89 GB usable) and 64-bit Operating System.

4. Resource of Networks

The network “NineNodes_net” is from “D. W. Hearn and J. Bouned Ribera, Flow Equilibrium by Penalty Methods, In: *Proceedings of the IEEE International Conference on Circuits and Computers*, Vol. 1, 1980, 162-164.”
The networks “Hull_net” and “NewHull_net” are from Dr. Toi.
All the other networks come from “The Traffic Assignment Problem” website (http://www.bgu.ac.il/~bargera/tntp/).

Anaheim_net.xlsx Austin_net.xlsx Barcelona_net.xlsx Braess_net.xlsx ChicagoRegional_net.xlsx ChicagoSketch_net.xlsx Hessen_net.xlsx Hull_net.xlsx

NewHull_net.xlsx NineNode_net.xlsx Philadelphia_net.xlsx PRISM_M_net.xlsx SiouxFalls_net.xlsx Terrassa_net.xlsx Winnipeg_net.xlsx

Appendix 1

```
 set
        k set of rows (links) /1*18/
        j set of parameters /1*6/
* 1:init node, 2:term node, 3:capacity, 4:free flow time, 5:B, 6:power

        n set of nodes /1*9/
        o set of zones /1*4/
        Alias (o,d);

parameter
        FirstNode
        data(k,j)
        od_data(o,d);

$CALL GDXXRW.EXE NineNode2_net.xls par=data rng=A7:G25 par=od_data rng=A27:E31
$GDXIN NineNode2_net.gdx
$LOAD data od_data
$GDXIN

FirstNode=1;

display data, od_data;

parameter
        a(n,k) whether the node i is the origin(destination) node of link k
        od(n,o,d)
        q(n,o);

        od(n,o,d)$(ord(o) ne ord(d))=od_data(o,d)$sameas(o,n)-od_data(o,d)$sameas(d,n);
        q(n,o)=sum(d,od(n,o,d));
```

```
*set the attribute for nodes
        a(n,k)$(ord(n) eq data(k,'1'))=1;
        a(n,k)$(ord(n) eq data(k,'2'))=-1;

parameter
        BC(k);
        BC(k)=data(k,'5')/(data(k,'3')**data(k,'6'));

display  BC;

variable
        Z the UE obj,
        TT_SO the SO obj
        ;

positive variable
        x(o,k) the flow of link k in od pair
        ;

equation flowcon(o,n) the node conservation,
        ocon(o,n,k) the o conservation,
        dcon(o,n,k) the d conservation,
        obj1 the UE obj,
        obj2 the SO obj
        ;

flowcon(o,n)..
        sum(k,a(n,k)*x(o,k)) =e= q(n,o);

ocon(o,n,k)$(sameas(n,o))..
        x(o,k)$(a(n,k) eq -1) =e= 0;
```

```
dcon(o,n,k)$((not(sameas(n,o))) and (ord(n) le FirstNode-1))..
    x(o,k)$(a(n,k) eq 1) =e= 0;

obj1..
    Z =e= sum(k,data(k,'4')*sum(o,x(o,k))
      +data(k,'4')*BC(k)*((sum(o,x(o,k))**(data(k,'6')+1))/(data(k,'6')+1)));

obj2..
    TT_SO =e= sum(k,data(k,'4')*sum(o,x(o,k))
      +data(k,'4')*BC(k)*(sum(o,x(o,k))**(data(k,'6')+1)));

OPTION LIMROW = 0, LIMCOL = 0, SOLPRINT = OFF, DECIMALS = 4
   SYSOUT = OFF, ITERLIM = 999999, RESLIM = 9999999, NLP=CONOPT;
model UE /flowcon, ocon, dcon, obj1/;
model SO /flowcon, ocon, dcon, obj2/;

solve UE using nlp minimizing Z;
Parameter
  f(k,*) the link flow of UE,
  TT_UE the total cost of UE
  ;

f(k,"I") = data(k,"1");
f(k,"J") = data(k,"2");
f(k,"flow") = sum(o,X.l(o,k));
TT_UE = sum(k,data(k,'4')*sum(o,x.l(o,k))
      +data(k,'4')*BC(k)*(sum(o,x.l(o,k))**(data(k,'6')+1)));

solve SO using nlp minimizing TT_SO;
Parameter
  fs(k,*) the link flow of SO
```

```
    ;

fs(k,"I") = data(k,"1");
fs(k,"J") = data(k,"2");
fs(k,"flow") = sum(o,X.l(o,k));

display f, fs, Z.L, TT_UE, TT_SO.L;
```